# 设定源文件文件夹
aux_source_directory(lib LIBTIN_SRC)

# 这个 GetOptWin 库可在此地址找到 https://gitee.com/yizhangss/getopt-win
if(WIN32)
	find_package(GetOptWin REQUIRED)
	message(STATUS "Found GetOpt-Win")
	include_directories(${GetOptWin_INC_DIR})
endif()

# 以下部分为库的编译
# 注意目标名必须唯一 所以不能直接生成相同名称的动态库与静态库
# 注意此处不必为目标名称添加lib前缀和相应后缀，cmake会自行添加
add_library(tin SHARED ${LIBTIN_SRC})
# 首先添加静态库的生成命令
add_library(tin_static STATIC ${LIBTIN_SRC})
# 设置静态库的输出名称从而获得与动态库名称相同的静态库
set_target_properties(tin_static PROPERTIES OUTPUT_NAME "tin")
# 设置输出目标属性以同时输出动态库与静态库
set_target_properties(tin PROPERTIES CLEAN_DIRECT_OUTPUT 1)
set_target_properties(tin_static PROPERTIES CLEAN_DIRECT_OUTPUT 1)
# 设置动态库的版本号
set_target_properties(tin PROPERTIES VERSION ${PROJECT_VERSION} SOVERSION ${PROJECT_VERSION_MAJOR}.${PROJECT_VERSION_MINOR})
# 设置库文件的输出地址
set(LIBRARY_OUTPUT_PATH ${PROJECT_BINARY_DIR}/lib)

# 设置编译选项
set(CMAKE_CXX_STANDARD 11)
set(CMAKE_CXX_FLAGS "${CMAKE_CXX_FLAGS} -O3")

set(CONFIG_FILE_PATH lib/cmake/${PROJECT_NAME})

configure_package_config_file(${PROJECT_SOURCE_DIR}/${PROJECT_NAME}Config.cmake.in 
	${CMAKE_BINARY_DIR}/${PROJECT_NAME}Config.cmake
	INSTALL_DESTINATION ${CONFIG_FILE_PATH}
	NO_CHECK_REQUIRED_COMPONENTS_MACRO)

write_basic_package_version_file(${CMAKE_BINARY_DIR}/${PROJECT_NAME}ConfigVersion.cmake
        VERSION ${PROJECT_VERSION}
        COMPATIBILITY SameMajorVersion)

# 库的安装命令
if(WIN32)
	install(TARGETS tin DESTINATION lib)
	install(TARGETS tin_static DESTINATION lib)
else()
	install(TARGETS tin tin_static
		EXPORT ${PROJECT_NAME}Targets
		LIBRARY DESTINATION lib
		ARCHIVE DESTINATION lib)
	install(FILES
		${CMAKE_BINARY_DIR}/${PROJECT_NAME}Config.cmake
		${CMAKE_BINARY_DIR}/${PROJECT_NAME}ConfigVersion.cmake
		DESTINATION ${CONFIG_FILE_PATH})
endif()
# 头文件安装命令
install(FILES lib/tin.h DESTINATION include)

# 以下部分为例子程序的编译
# 设置可执行文件的输出地址
set(EXECUTABLE_OUTPUT_PATH ${PROJECT_BINARY_DIR}/bin)

# 例子的编译方法
macro(add_demo name)
	# 添加可执行文件 命令行
	add_executable(${name} demo/${name}.cpp)
	# 为安装文件添加动态库的搜索地址 在Windows下并没有什么用 直接忽略
	set_target_properties(${name} PROPERTIES INSTALL_RPATH /usr/local/lib)
	# 链接动态库
	target_link_libraries(${name} PUBLIC tin)
endmacro()

add_demo(demo)
add_demo(demo2)
add_demo(demo3)
add_demo(demo4)

# 工具的编译方法
macro(add_tool name)
	# 添加可执行文件 命令行
	add_executable(${name} tool/${name}.cpp)
	# 为安装文件添加动态库的搜索地址 在Windows下并没有什么用 直接忽略
	set_target_properties(${name} PROPERTIES INSTALL_RPATH /usr/local/lib)
	# 链接动态库
	target_link_libraries(${name} PUBLIC tin)
	# 安装命令
	install(TARGETS ${name} DESTINATION sbin)
endmacro()

add_tool(grd2tin)
add_tool(rnd2tin)
# Windows下需要单独链接getopt库 这里有点小BUG 可能不会在VStudio中正确添加附件依赖项路径 此时可手动添加getopt的动态库地址
if(WIN32)
	link_directories(${GetOptWin_LIB_DIR})
	target_link_libraries(grd2tin PUBLIC ${GetOptWin_LIB})
	target_link_libraries(rnd2tin PUBLIC ${GetOptWin_LIB})
endif()